
1 Overview
This tutorial demonstrates how to set up and perform analyses using common nucleotide substitution
models. The substitution models used in molecular evolution are continuous time Markov models, which
are fully characterized by their instantaneous-rate matrix:

Q =


−µA µGA µCA µTA
µAG −µG µCG µTG
µAC µGC −µC µTC
µAT µGT µCT −µT

 ,

where µij represents the instantaneous rate of substitution from state i to state j. Given the instantaneous-
rate matrix, Q, we can compute the corresponding transition probabilities for a branch of length t, P (t),
by exponentiating the rate matrix:

P (t) =


pAA(t) pGA(t) pCA(t) pTA(t)
pAG(t) pGG(t) pCG(t) pTG(t)
pAC(t) pGC(t) pCC(t) pTC(t)
pAT (t) pGT (t) pCT (t) pTT (t)

 = eQt =
∞∑
j=0

tQj

j! .

Each specific substitution model has a uniquely defined instantaneous-rate matrix, Q.

In this tutorial you will perform phylogeny inference under common models of DNA sequence evolution:
JC, F81, HKY85, GTR, GTR+Gamma and GTR+Gamma+I. For all of these substitution models, you
will perform an MCMC analysis to estimate phylogeny and other model parameters.

Requirements

We assume that you have read and hopefully completed the following tutorials:

• RB_Getting_Started

• RB_Basics_Tutorial

2 Data and files
We provide several data files which we will use in this tutorial. You may want to use your own data instead.
In the data folder, you will find the following files

• primates_cytb.nex: Alignment of the cytochrome b subunit from 23 primates representing 14 of
the 16 families (Indriidae and Callitrichidae are missing).

3 Example: Character Evolution under the Jukes-Cantor Substitution
Model

3.1 Getting Started

The first section of this exercise involves: (1) setting up a Jukes-Cantor (JC) substitution model for an
alignment of the cytochrome b subunit; (2) approximating the posterior probability of the tree topology
and branch lengths (and all other parameters) using MCMC, and; (3) summarizing the MCMC output by
computing the maximum a posteriori tree.

root internal tip

SijSijSij

Sp̃(i)j Sp̃(i)j

j 2 N

i = root i 2 internals i 2 tips

Q

⌫i ⌫i

�

 

Figure 1: Graphical model representation of a simple phylogenetic model. The graphical model shows the depen-
dencies between the parameters. Here, the rate matrix Q is constant variable because it is fixed and does not depend
on any parameters. The only free parameters of this model, the unconstrained Jukes-Cantor model, are the tree
topology Ψ and the branch lengths νi.

The general structure of the model is represented in Figure 1. We first consider the simplest substitution
model described by Jukes and Cantor (1969). The instantaneous-rate matrix for the JC substitution model
is defined as

QJC69 =


∗ 1 1 1
1 ∗ 1 1
1 1 ∗ 1
1 1 1 ∗

 ,

which has the advantage that the transition probability matrix can be computed analytically:

PJC69 =

In the later exercises you will be asked to specify more complex substitution models. Don’t be scared
by the math! RevBayes will take care of all the computations for you. Here we only provide some of the
equations for the models in case you might be interested in the details. You will be able to complete the
exercises without understanding the underlying math.

The files for this example analysis are provided for you, which can easily be run using the source()
function in the RevBayes console:

source("RevBayes_scripts/JukesCantor.Rev")

If everything loaded properly, then you should see the program initiate the Markov chain Monte Carlo
analysis that estimates the posterior distribution. If you continue to let this run, then you will see it
output the states of the Markov chain once the MCMC analysis begins. (It is worth noting, however, that
the file JukesCantor.Rev performs shorter runs with fewer generations for the purposes of illustration.)

Ultimately, this is how you will execute most analyses in RevBayes , with the full specification of the model
and analyses contained in the sourced files. You could easily run this entire analysis on your own data
by substituting your data file name for that in the model-specification file. However, it is important to
understand the components of the model to be able to take full advantage of the flexibility and richness of
RevBayes. Furthermore, without inspecting the Rev scripts sourced in JukesCantor.Rev, you may end up
inadvertently performing inappropriate analyses on your dataset, which would be a waste of your time and
CPU cycles. The next steps will walk you through the full specification of the model and MCMC analyses.

3.2 Loading the Data

Download data and output files (if you don’t have them already) from: http://revbayes.github.io/tutorials.html→

First load in the sequences using the readDiscreteCharacterData() function.

data <- readDiscreteCharacterData("data/primates_cytb.nex")

Executing these lines initializes the data matrix as the respective Rev variables. To report the current
value of any variable, simply type the variable name and press enter. For the data matrix, this provides
information about the alignment:

data
DNA character matrix with 23 taxa and 1141 characters
=====================================================
Origination: primates_cytb.nex
Number of taxa: 23
Number of included taxa: 23
Number of characters: 1141
Number of included characters: 1141
Datatype: DNA

Next we will specify some useful variables based on our dataset. The variable data has member functions
that we can use to retrieve information about the dataset. These include the number of species (n_species),
the tip labels (names), and the number of internal branches (n_branches). Each of these variables will be
necessary for setting up different parts of our model.

n_species <- data.ntaxa()
names <- data.names()
n_branches <- 2 * n_species - 3

Additionally, we set up a counter variable for the number of moves that we already added to our analysis.
[Recall that moves are algorithms used to propose new parameter values during the MCMC simulation.]
This will make it much easier if we extend the model or analysis to include additional moves or to remove
some moves.

mi = 0

You may have noticed that we used the = operator to create the move index. This simply means that the
variable is not part of the model. You will later see that we use this operator more often, e.g., when we
create moves and monitors.

With the data loaded, we can now proceed to specify our Jukes-Cantor substitution model.

3.3 Jukes-Cantor Substitution Model

A given substitution model is defined by its corresponding instantaneous-rate matrix, Q. The Jukes-Cantor
substitution model does not have any free parameters (as the substitution rates are all assumed to be equal),
so we can define it as a constant variable. The function fnJC(n) will create an instantaneous-rate matrix
for character with n states. Since we use DNA data here, we create a 4x4 instantaneous-rate matrix:

Q <- fnJC(4)

You can see the rates of the Q matrix by typing

Q
[ [ -1.0000, 0.3333, 0.3333, 0.3333 ] ,

0.3333, -1.0000, 0.3333, 0.3333 ] ,
0.3333, 0.3333, -1.0000, 0.3333 ] ,
0.3333, 0.3333, 0.3333, -1.0000 ] ]

As you can see, all substitution rates are equal.

3.4 Tree Topology and Branch Lengths

The tree topology and branch lengths are stochastic nodes in our phylogenetic model. In Figure 1, the
tree topology is denoted Ψ and the length of the branch leading to node i is νi.

We will assume that all possible labeled, unrooted tree topologies have equal probability. This is the
dnUniformTopology() distribution in RevBayes. Specify the topology stochastic node by passing in the
tip labels names to the dnUniformTopology() distribution:

topology ~ dnUniformTopology(names)

Some types of stochastic nodes can be updated by a number of alternative moves. Different moves may
explore parameter space in different ways, and it is possible to use multiple different moves for a given
parameter to improve mixing (the efficiency of the MCMC simulation). In the case of our unrooted tree
topology, for example, we can use both a nearest-neighbor interchange move (mvNNI) and a subtree-prune
and regrafting move (mvSPR). These moves do not have tuning parameters associated with them, thus you
only need to pass in the topology node and proposal weight

moves[++mi] = mvNNI(topology, weight=1.0)
moves[++mi] = mvSPR(topology, weight=1.0)

The weight specifies how often the move will be applied either on average per iteration or relative to all
other moves. Have a look at the MCMC tutorial for more details about moves and MCMC strategies:
http://revbayes.github.io/tutorials.html

Next we have to create a stochastic node for each of the 2N−3 branches in our tree (whereN = n_species).
We can do this using a for loop — this is a plate in our graphical model. In this loop, we can create each
of the branch-length nodes and assign each move. Copy this entire block of Rev code into the console:

for (i in 1:n_branches) {
br_lens[i] ~ dnExponential(10.0)
moves[++mi] = mvScale(br_lens[i])

}

It is convenient for monitoring purposes to add the tree length as deterministic variable. The tree length
is simply the sum of all branch lengths. Accordingly, the tree length can be computed using the sum()
function, which calculates the sum of any vector of values.

TL := sum(br_lens)

Finally, we can create a phylogram (a phylogeny in which the branch lengths are proportional to the
expected number of substitutions/site) by combining the tree topology and branch lengths. We do this
using the treeAssembly() function, which applies the value of the ith member of the br_lens vector to
the branch leading to the ith node in topology. Thus, the phylogeny variable is a deterministic node:

phylogeny := treeAssembly(topology, br_lens)

3.5 Putting it All Together

We have fully specified all of the parameters of our phylogenetic model—the tree topology with branch
lengths, and the substitution model that describes how the sequence data evolved over the tree with
branch lengths. Collectively, these parameters comprise a distribution called the phylogenetic continuous-
time Markov chain, and we use the PhyloCTMC constructor function to create this node. This distribution
requires several input arguments: (1) the tree with branch lengths; (2) the instantaneous-rate matrix Q,
and; (3) the type of character data.

seq ~ dnPhyloCTMC(tree=phylogeny, Q=Q, type="DNA")

Once the PhyloCTMC model has been created, we can attach our sequence data to the tip nodes in the tree.

seq.clamp(data)

[Note that although we assume that our sequence data are random variables—they are realizations of
our phylogenetic model—for the purposes of inference, we assume that the sequence data are “clamped”.]
When this function is called, RevBayes sets each of the stochastic nodes representing the tips of the tree
to the corresponding nucleotide sequence in the alignment. This essentially tells the program that we have
observed data for the sequences at the tips.

Finally, we wrap the entire model to provide convenient access to the DAG. To do this, we only need to
give the model() function a single node. With this node, the model() function can find all of the other
nodes by following the arrows in the graphical model:

mymodel = model(Q)

Now we have specified a simple phylogenetic analysis—each parameter of the model will be estimated from
every site in our alignment. If we inspect the contents of mymodel we can review all of the nodes in the
DAG:

mymodel

3.6 Performing an MCMC Analysis Under the Uniform Model

In this section, will describe how to set up the MCMC sampler and summarize the resulting posterior
distribution of trees.

3.6.1 Specifying Monitors

For our MCMC analysis, we need to set up a vector of monitors to record the states of our Markov chain.
The monitor functions are all called mn*, where * is the wildcard representing the monitor type. First, we
will initialize the model monitor using the mnModel function. This creates a new monitor variable that will
output the states for all model parameters when passed into a MCMC function.

monitors[1] = mnModel(filename="output/primates_cytb_JC_posterior.log",printgen=10,
separator = TAB)

The mnFile monitor will record the states for only the parameters passed in as arguments. We use this
monitor to specify the output for our sampled trees and branch lengths.

monitors[2] = mnFile(filename="output/primates_cytb_JC_posterior.trees",printgen=10,
separator = TAB, phylogeny)

Finally, create a screen monitor that will report the states of specified variables to the screen with mnScreen:

monitors[3] = mnScreen(printgen=1000, TL)

3.6.2 Initializing and Running the MCMC Simulation

With a fully specified model, a set of monitors, and a set of moves, we can now set up the MCMC algorithm
that will sample parameter values in proportion to their posterior probability. The mcmc() function will
create our MCMC object:

mymcmc = mcmc(mymodel, monitors, moves)

We may wish to run the .burnin() member function. Recall that this function does not specify the
number of states that we wish to discard from the MCMC analysis as burnin (i.e., the samples collected
before the chain converges to the stationary distribution). Instead, the .burnin() function specifies a
completely separate preliminary MCMC analysis that is used to tune the scale of the moves to improve
mixing of the MCMC analysis.

mymcmc.burnin(generations=10000,tuningInterval=1000)

Now, run the MCMC:

mymcmc.run(generations=30000)

When the analysis is complete, you will have the monitored files in your output directory.

Methods for visualizing the marginal densities of parameter values are not currently available in RevBayes it-
self. Thus, it is important to use programs like Tracer (Rambaut and Drummond 2011) to evaluate mixing
and non-convergence. (RevBayes does, however, have a tool for convergence assessment called beca.)

Look at the file called output/primates_cytb_JC_posterior.log in Tracer.→

3.7 Exercise 1

We are interested in the phylogenetic relationship of the Tarsiers. Therefore, we need to summarize the
trees sampled from the posterior distribution. RevBayes can summarize the sampled trees by reading in
the tree-trace file:

treetrace = readTreeTrace("output/primates_cytb_JC_posterior.trees", treetype="non-
clock")

treetrace.summarize()

The mapTree() function will summarize the tree samples and write the maximum a posteriori tree to file:

mapTree(treetrace,"output/primates_cytb_JC.tree")

Fill in the following table as you go through the tutorial.

Table 1: Posterior probabilities of phylogenetic relationship∗.
Model Lemuroidea Lorisoidea Platyrrhini Catarrhini other

Jukes-Cantor

HKY85

F81

GTR

GTR+Γ

GTR+Γ+I

Your model 1

Your model 2

Your model 3
∗you can edit this table

4 The Hasegawa-Kishino-Yano (HKY) 1985 Substitution Model
The Jukes-Cantor model assumes that all substitution rates are equal, which also implies that the stationary
frequencies of the four nucleotide bases are equal. These assumptions are not very biologically reasonable,
so we might wish to consider a more realistic substitution model that relaxes some of these assumptions.
For example, we might allow stationary frequencies, π, to be unequal, and allow rates of transition and
transversion substitutions to differ, κ. This corresponds to the substitution model proposed by Hasegawa
et al. (1985; HKY), which is specified with the following instantaneous-rate matrix:

QHKY =


∗ κπC πG πT
κπA ∗ πC πT
πA πC ∗ κπT
πA πC κπG ∗

 .

Use the file JukesCantor.Rev as a starting point for the HKY analysis.→

Note that we are adding two new variables to our model. We can define a variable pi for the stationary
frequencies that are drawn from a flat Dirichlet distribution by

pi_prior <- v(1,1,1,1)
pi ~ dnDirichlet(pi_prior)

Since pi is a stochastic variable, we need to specify a move to propose updates to it. A good move on
variables drawn from a Dirichlet distribution is the mvSimplexElementScale. This move randomly takes
an element from the simplex, proposes a new value for it drawn from a Beta distribution, and then rescales
all values of the simplex to sum to 1 again.

moves[++mi] = mvSimplexElementScale(pi, weight=4.0)

The second new variable is κ, which specifies the ratio of transition-transversion rates. The κ parameter
must be a positive-real number and a natural choice as the prior distribution is the lognormal distribution:

kappa ~ dnLnorm(0.0,1.25)

Again, we need to specify a move for this new stochastic variable. A simple scaling move should do the
job.

moves[++mi] = mvScale(kappa, weight=1.0)

Finally, we need to create the HKY instantaneous-rate matrix using the fnHKY function:

Q := fnHKY(kappa,pi)

This should be all for the HKY model.

Don’t forget to change the output file names, otherwise your old analyses files will be overwritten.→

4.1 Exercise 2

• Copy the file called JukesCantor.Rev and modify it by including the necessary parameters to specify
the HKY substitution model.

• Run an MCMC analysis to estimate the posterior distribution under the HKY substitution model.

• Are the resulting estimates of the base frequencies equal? If not, how much do they differ? Are the
estimated base frequencies similar to the empirical base frequencies? The empirical base frequencies
are the frequencies of the characters in the alignment, which can be computed with RevBayes by
data.empiricalBaseFrequencies().

• Is the inferred rate of transition substitutions higher than the rate of transversion substitutions? If
so, by how much?

• Like the HKY model, the Felsenstein 1981 (F81)substitution model has unequal stationary frequen-
cies, but it assumes equal transition-transversion rates (Felsenstein 1981). Can you set up the F81
model and run an analysis?

• Complete the table of the phylogenetic relationship of Tarsiers.

5 The General Time-Reversible (GTR) Substitution Model
The HKY substitution model can accommodate unequal base frequencies and different rates of transition
and transversion substitutions. Despite these extensions, the HKY model may still be too simplistic for
many real datasets. Here, we extend the HKY model to specify the General Time Reversible (GTR)
substitution model (Tavaré 1986), which allows all six exchangeability rates to differ (Figure 2). The
instantaneous-rate matrix for the GTR substitution model is

QGTR =


−(x1 + x2 + x3) π1x1

π2
π1x2
π3

π1x3
π4

x1 −(π1x1
π2

+ x4 + x5) π2x4
π3

π2x5
π4

x2 x4 −(π1x2
π3

+ π2x4
π3

+ x6) π3x6
π4

x3 x5 x6 −(π1x3
π4

+ π3x6
π4

Figure 2: Graphical model representation of the General Time Reversible (GTR) phylogenetic model.

5.1 Exchangeability Rate Parameters

The GTR model requires that we define and specify a prior on the six exchangeability rates, which we will
describe using a flat Dirichlet distribution. As we did previously for the Dirichlet prior on base frequencies,
we first define a constant node specifying the vector of concentration-parameter values using the v()
function:

er_prior <- v(1,1,1,1,1,1)

This node defines the concentration-parameter values of the Dirichlet prior distribution on the exchange-
ability rates. Now, we can create a stochastic node for the exchangeability rates using the dnDirichlet()
function, which takes the vector of concentration-parameter values as an argument and the ~ operator.
Together, these create a stochastic node named er (θ in Figure 2):

er ~ dnDirichlet(er_prior)

The Dirichlet distribution assigns probability densities to a group of parameters: e.g., those that measure
proportions and must sum to 1. Here, we have specified a six-parameter Dirichlet prior, where each value
describes one of the six relative rates of the GTR model: (1) A� C; (2) A� G; (3) A� T ; (4) C � G;
(5) C � T ; (6) G� T . The input parameters of a Dirichlet distribution are called shape (or concentration)
parameters. The expectation and variance for each variable are related to the sum of the shape parameters.
The prior we specified above is a ‘flat’ or symmetric Dirichlet distribution; all of the shape parameters
are equal (1,1,1,1,1,1). This describes a model that allows for equal rates of change between nucleotides,
such that the expected rate for each is equal to 1

6 (Figure 3a). We might also parameterize the Dirichlet
distribution such that all of the shape parameters were equal to 100, which would also specify a prior with
an expectation of equal exchangeability rates (Figure 3b). However, by increasing the values of the shape
parameters, er_prior <- v(100,100,100,100,100,100), the Dirichlet distribution will more strongly
favor equal exchangeability rates; (i.e., providing is a relatively informative prior). Alternatively, we might
consider an asymmetric Dirichlet parameterization that could reflect a strong prior belief that transition
and transversion substitutions occur at different rates. For example, we might specify the prior density
er_prior <- v(4,8,4,4,8,4). Under this model, the expected rate for transversions would be 4

32 and
that for transitions would be 8

32 , and there would be greater prior probability on sets of GTR rates that
matched this configuration (Figure 3c). Yet another aymmetric prior could specify that each of the six
GTR rates had a different value conforming to a Dirichlet(2,4,6,8,10,12). This would lead to a different
prior probability density for each rate parameter (Figure 3d). Without strong prior knowledge about the
pattern of relative rates, however, we can better reflect our uncertainty by using a vague prior on the
GTR rates. Notably, all patterns of relative rates have the same probability density under er_prior <-
v(1,1,1,1,1,1).

For each stochastic node in our model, we must also specify a proposal mechanism if we wish to estimate
that parameter. The Dirichlet prior on our parameter er creates a simplex of values that sum to 1.

moves[++mi] <- mvSimplexElementScale(er, weight=3)

We can use the same type of distribution as a prior on the 4 stationary frequencies (πA, πC , πG, πT ) since
these parameters also represent proportions. Specify a flat Dirichlet prior density on the base frequencies:

pi_prior <- v(1,1,1,1)
pi ~ dnDirichlet(pi_prior)

The node pi represents the π node in Figure 2. Now add the simplex scale move on the stationary
frequencies to the moves vector:

12

Figure 3: Four different examples of Dirichlet priors on exchangeability rates.

moves[++mi] <- mvSimplexElementScale(pi, weight=2)

We can finish setting up this part of the model by creating a deterministic node for the GTR instantaneous-
rate matrix Q. The fnGTR() function takes a set of exchangeability rates and a set of base frequencies to
compute the instantaneous-rate matrix used when calculating the likelihood of our model.

Q := fnGTR(er,pi)

5.2 Execise 3

• Use one of your previous analysis files—either the JukesCantor.Rev or HKY.Rev—to specify a GTR
analysis in a new file called GTR.Rev. Adapt the old analysis to be performed under the GTR
substitution model.

• Run an MCMC analysis to estimate the posterior distribution.

• Complete the table of the phylogenetic relationship of Tarsiers.

6 The Discrete Gamma Model of Among Site Rate Variation
Members of the GTR family of substitution models assume that rates are homogeneous across sites,
an assumption that is often violated by real data. We can accommodate variation in substitution rate
among sites (ASRV) by adopting the discrete-gamma model (Yang 1994). This model assumes that the
substitution rate at each site is a random variable that is described by a discretized gamma distribution,
which has two parameters: the shape parameter, α, and the rate parameter, β. In order that we can
interpret the branch lengths as the expected number of substitutions per site, this model assumes that the
mean site rate is equal to 1. The mean of the gamma is equal to α/β, so a mean-one gamma is specified
by setting the two parameters to be equal, α = β. This means that we can fully describe the gamma
distribution with the single shape parameter, α. The degree of among-site substitution rate variation is
inversely proportional to the value of the α-shape parameter. As the value of the α-shape increases, the
gamma distribution increasingly resembles a normal distribution with decreasing variance, which therefore
corresponds to decreasing levels of ASRV (Figure 4). By contrast, when the value of the α-shape parameter
is < 1, the gamma distribution assumes a concave distribution that concentrates most of the prior density
on low rates, but retains some prior mass on sites with very high rates, which therefore corresponds to high
levels of ASRV (Figure 4). Note that, when α = 1, the gamma distribution collapses to an exponential
distribution with a rate parameter equal to β.

Figure 4: The probability density of mean-one gamma-distributed rates for different values of the α-shape parameter.

We typically lack prior knowledge regarding the degree of ASRV for a given alignment. Accordingly, rather
than specifying a precise value of α, we can instead estimate the value of the α-shape parameter from the
data. This requires that we specify a diffuse (relatively ‘uninformative’) prior on the α-shape parameter.
For this analysis, we will use an exponential distribution with a rate parameter, shape_prior, equal to
0.05. An exponential prior assigns non-zero probability on values of α ranging from 0 to ∞. The rate
parameter of an exponential distribution, often denoted λ, controls both the mean and variance of this
distribution, such that the expected (or mean) value of α is: E[α] = 1

λ . Thus, if we set λ = 0.05, then
E[α] = 20.

This approach for accommodating ASRV is another example of a hierarchical model (Figure 5). That is,
variation in substitution rates across sites is addressed by applyig a site-specific rate multiplier to each of
the j sites, rj . These rate-multipliers are drawn from a discrete, mean-one gamma distribution; the shape
of this prior distribution (and the corresponding degree of ASRV) is governed by the α-shape parameter.

The α-shape parameter, in turn, is treated as an exponentially distributed random variable. Finally, the
shape of the exponential prior is governed by the rate parameter, λ, which is set to a fixed value.

Figure 5: Graphical model representation of the General Time Reversible (GTR) + Gamma phylogenetic model.

6.1 Setting up the Gamma Model in RevBayes

Create a constant node called shape_prior for the rate parameter of the exponential prior on the gamma-
shape parameter (this is represented as the constant λ-rate parameter in Figure 5):

shape_prior <- 0.05

Then create a stochastic node called alpha with an exponential prior (this represents the stochastic node
for the α-shape parameter in Figure 5):

alpha ~ dnExponential(shape_prior)

The way the ASRV model is implemented involves discretizing the mean-one gamma distribution into a
set number of rate categories, k. Thus, we can analytically marginalize over the uncertainty in the rate at
each site. The likelihood of each site is averaged over the k rate categories, where the rate multiplier is
the mean (or median) of each of the discrete k categories. To specify this, we need a deterministic node
that is a vector that will hold the set of k rates drawn from the gamma distribution with k rate categories.
The fnDiscretizeGamma() function returns this deterministic node and takes three arguments: the shape
and rate of the gamma distribution and the number of categories. Since we want to discretize a mean-one
gamma distribution, we can pass in alpha for both the shape and rate.

Initialize the gamma_rates deterministic node vector using the fnDiscretizeGamma() function with 4 bins:

gamma_rates := fnDiscretizeGamma( alpha, alpha, 4 )

Note that here, by convention, we set k = 4. The random variable that controls the rate variation is the
stochastic node alpha. We will apply a simple scale move to this parameter.

moves[++mi] = mvScale(alpha, weight=2.0)

Remember that you need to call the PhyloCTMC constructor to include the new site-rate parameter:

seq ~ dnPhyloCTMC(tree=phylogeny, Q=Q, siteRates=gamma_rates , type="DNA")

6.2 Execise 4

Modify the previous GTR analysis to specify the GTR+Gamma model. Run an MCMC simulation to
estimate the posterior distribution.

• Is there an impact on the estimated phylogeny compared with the previous analyses? Look at the
MAP tree and the posterior probabilities of the clades.

• What is the estimated tree length? Is the estimate different to the previous analysis? What could
cause this?

• Complete the table of the phylogenetic relationship of Tarsiers.

7 Modeling Invariable Sites
All of the substitution models described so far assume that the sequence data are potentially variable.
That is, we assume that the sequence data are random variables; specifically, we assume that they are
realizations of the specified PhyloCTMC distribution. However, some sites may not be free to vary—when
the substitution rate of a site is zero, it is said to be invariable. Invariable sites are often confused with
invariant sites—when each species exhibits the same state, it is said to be invariant. The concepts are
related but distinct. If a site is truly invariable, it will necessarily give rise to an invariant site pattern, as
such sites will always have a zero substitution rate. However, an invariant site pattern may be achieved
via multiple substitutions that happen to end in the same state for every species.

Here we describe an extension to our phylogenetic model to accommodate invariable sites. Under the
invariable-sites model (Hasegawa et al. 1985), each site is invariable with probability pinvar, and variable
with probability 1−pinvar.

First, let’s have a look at the data and see how many invariant sites we have:

data.getNumInvariantSites()

There seem to be a substantial number of invariant sites.

Now let’s specify the invariable-sites model in RevBayes. We need to specify the prior probability that a
site is invariable. A Beta distribution is a common choice for parameters representing probabilities.

pinvar ~ dnBeta(1,1)

The Beta(1,1) distribution is a flat prior distribution that specifies equal probability for all values between
0 and 1.

Then, as usual, we add a move to change this stochastic variable; we’ll used a simple sliding window move.

moves[mi++] = mvSlide(pinvar)

7.1 Exercise 5

• Extend the GTR model to account for invariable sites and run an analysis.

• What is the estimated probability of invariable sites and how does it relate to the ratio of invariant
sites to the total number of sites?

• Extend the GTR+Γ model to account for invariable sites and run an analysis.

• What is the estimated probability of invariable sites now?

• Complete the table of the phylogenetic relationship of Tarsiers.

Questions about this tutorial can be directed to:
• Tracy Heath (email: tracyh@berkeley.edu)
• Michael Landis (email: mlandis@berkeley.edu)
• Sebastian Höhna (email: sebastian.hoehna@gmail.com)
• Brian R. Moore (email: brianmoore@ucdavis.edu)

